Merging Data Seven Different Ways
David Franklin, Independent Consultant, New Hampshire, USA

ABSTRACT
Merging data is one of the fundamental functions carried out when manipulating data to bring it into a form for either storage or analysis. The use of the MERGE statement inside a datastep is the most common way this task is done within the SAS language but there are others. This paper looks at the seven possible methods, including the use of the MERGE statement, for a one to one, or one to many merge, introducing the SAS® code needed to combine the data. No one method is better than another but some pointers will be given on choosing a method for your data.

INTRODUCTION
Merging variables from one dataset into another is one of the basic data manipulation tasks that a SAS programmer has to do. The most common way to merge on data is using the MERGE statement in the DATA step but there are six other ways that can help. First though, some data:

Dataset: PATDATA
SUBJECT  TRT_CODE
124263      A
124264      A
124265      B
124266      B

Dataset: ADVERSE
SUBJECT  EVENT
124263   HEADACHE
124266   FEVER
124266   NAUSEA
124267   FRACTURE

This data will be used throughout the paper for each method described.

MERGING THE DATA

MERGE IN A DATA STEP
The most commonly used statement used when merging data within SAS is the MERGE statement used inside a datastep, an example of which is given below:

```
DATA alldata0;
    MERGE adverse (in=a) patdata (in=b);
    BY subject;
    IF a;
RUN;
```

This method is the most common way of merging data as it gives control the the way data is to be merged. In the example, the ADVERSE and PATDATA records are merged by SUBJECT and those records that come from the ADVERSE dataset are output - this results in subject 124267 not having a TRT_CODE value in the ALLDATA0 dataset, and subjects 124264 and 124265 not being represented in the same. Most commonly the datastep is preceded by a call to the SORT procedure making sure that both ADVERSE and PATDATA are in the same sort order, but it is possible to use indexed datasets instead but the indexes must be defined before the datastep that does the merging of the data.

MERGE WITH SQL
SAS version 6.07 introduced SQL into SAS which gave the ability to merge data using the SQL language. To merge our two datasets and get the same result as in the code above, the SQL code would look something similar
to that below:

```sql
PROC SQL;
  CREATE TABLE alldata0 AS
    SELECT a.*, b.trt_code
    FROM adverse a
    LEFT JOIN
    patdata b
    ON a.subject=b.subject;
  QUIT;
RUN;
```

SQL is a well known language that is very good at working with databases and is liked by many who deal with
large datasets.

**MERGE WITH SET-KEY**

Over the years many options have been added to the SET statement which brings the third method for merging
data, using the KEY= option as shown in the following example:

```sas
DATA alldata0;
  SET adverse;
  SET patdata KEY=subject /UNIQUE;
  DO;
    IF _IORC_ THEN DO;
      _ERROR_=0;
      trt_code=''
    END;
  END;
RUN;
```

Before the third example is run the dataset PATDATA must have an index created inside it, using either the INDEX
statement inside a DATASETS or SQL procedure, or INDEX option inside a DATA step. It is important to have the
DO loop is if no match is found then TRT_CODE will be set to missing - if this is not done then unexpected results
may occur.

**MERGE WITH FORMAT**

The fourth method that is useful creates a format from the PATDATA dataset and and sets the treatment from the
created format:

```sas
DATA fmt;
  RETAIN fmtname 'TRT_FMT' type 'C';
  SET patdata;
  RENAME subject=start trt_code=label;
RUN;
PROC FORMAT CNTLIN=fmt;
RUN;
DATA alldata0;
  SET adverse;
  ATTRIB trt_code LENGTH=$1 LABEL='Treatment Code';
  trt_code=PUT(subject,$trt_fmt.);
RUN;
```

In the example a character format TRT_FMT is created from the PATDATA dataset, and then this format is used to
set the TRT_CODE variable within the ADVERSE dataset. This method is useful as the data does not have to be
sorted or indexed beforehand.

**MERGE WITH HASH TABLE**

Since version 9.1 another possibility that has been available is the use of hash tables. Many papers have been
written about this recent feature, how it works, and their use within SAS - references to some notable papers are
in the Reference section below. The code below does the merge required:
DATA alldata0;
  IF _n_=0 THEN SET patdata;
  IF _n_=1 THEN DO;
    DECLARE HASH _h1
      (dataset: "PATDATA");
    rc=_h1.definekey("SUBJECT");
    rc=_h1.definedata("TRT_CODE");
    rc=_h1.definedone();
    call missing(SUBJECT,TRT_CODE);
  END;
  SET adverse;
  rc=_h1.find();
  IF rc^=0 THEN trt_code=" ";
  DROP rc;;
RUN;

In the example above, the dataset PATDATA gets loaded into a hash table, then the ADVERSE dataset is loaded into the datastep and the match is made using the FIND() method.

MERGE WITH ARRAY

A variation on the hash table is to load the dataset with unique records into an array and then do the match, as the following example demonstrates:

DATA _null_;  SET sashelp.vtable;
WHERE libname='WORK';  WHERE ALSO memname in('PATDATA','ADVERSE');
CALL SYMPUT('X'||memname,put(nobs,8.));
RUN;
DATA alldata0;
  LENGTH trt_code $1;
  ARRAY f{&xpatdata.,2} $6 _TEMPORARY_;
  DO i=1 TO &xpatdata.;
    SET patdata (RENAME=(trt_code=trt_code_dict));
    f{i,1}=PUT(subject,6.);
    f{i,2}=trt_code_dict;
  END;
  DO i=1 TO &xadverse.;
    SET adverse;
    trt_code='';
    DO j=1 TO &xpatdata.;
      IF subject=INPUT(f{j,1},best.) THEN DO;
        trt_code=f{j,2};
        OUTPUT;
      END;
      IF ^MISSING(trt_code) THEN LEAVE;
    END;
    IF MISSING(trt_code) THEN OUTPUT;
  END;
  DROP i j trt_code_dict;
RUN;

The first datastep finds the number of records within PATDATA and ADVERSE so that the correct number of elements can be set for the array and the correct number of iterations is used when calling the ADVERSE dataset. The method above does have one surprising feature - the line:

  IF subject=INPUT(f{j,1},best.) THEN DO;

where the actual compare is done, can be changed to use any comparison, whether it be an INDEX function or greater than/less than operators.
MERGE WITH MODIFY

The last merge considered uses the MODIFY statement – this is an interesting technique as it is necessary to do a loop within a loop due to the ADVERSE dataset having multiple records per subject:

```sas
DATA adverse;
   DO p = 1 TO totobs;
      _IORC_ = 0;
      SET patdata POINT=p NOBS=totobs;
      DO WHILE(_IORC_=%SYSRC(_SOK));
         MODIFY adverse KEY=subject;
         SELECT (_IORC_);
         WHEN (%SYSRC(_SOK)) DO;              /*MATCH FOUND*/
            SET patdata POINT=p; trtc=trtcode; REPLACE;
         END;
         WHEN (%SYSRC(_DSENOM)) _ERROR_ = 0;  /*NO MATCH FOUND*/
         END;
         OTHERWISE DO;                        /*A MAJOR PROBLEM SOMEWHERE*/
            PUT 'ERROR: _IORC_ = ' _IORC_ / 'PROGRAM HALTED.';
            _ERROR_ = 0; STOP;
         END;
      END;
   END;
STOP;
RUN;
```

It is necessary to note that this method modifies the existing ADVERSE dataset and will not create the ALLDATA0 dataset as the previous methods gave. Note also that the dataset ADVERSE has an index called SUBJECT applied before the datastep is run.

CONCLUSION

As shown in the paper there are a number of methods which can be used to merge data, beyond the MERGE statement within a DATA step. No one method is better than another, and the methods shown here are by no means exhaustive. It is only though trying these different methods at your site that you will see resource efficiencies between the methods.

REFERENCES


CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

David Franklin
16 Roberts Road
Litchfield, NH 03052
Tel/Fax: 603-262-9160
Email: 100316.3451@compuserve.com
Web: http://ourworld.compuserve.com/homepages/dfranklinuk

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies.